

TransLucid: Higher-order Indexical ISWIM

John Plaice*

Version 1, 21 June 2020

Version 2, 9 July 2020

Abstract

We present a higher-order functional language in which variables define arbitrary-dimensional entities, and a multidimensional runtime context is used to index the variables. We give an intuitive presentation of the language, present the denotational semantics, and demonstrate how function applications over these potentially infinite data structures can be transformed into manipulations of the runtime context.

1 Introduction

We present a functional language called TransLucid defining arbitrary-dimensional variables, which are evaluated in a lazy manner. The indexing of these variables is done through a runtime multidimensional context, an unordered set of $(dimension, ordinate)$ pairs. Structuring of expressions is done with **where** clauses, in which one may introduce both dimension identifiers and variable identifiers. To simplify the presentation, we have chosen to make TransLucid untyped; all results presented here generalize naturally to the typed situation.

The paper will consist of two main parts: (1) an informal introduction to the language, through the presentation of more and more complex examples; and (2) a formal presentation of the language, with both syntax and denotational semantics.

Variables and expressions are indexed by the runtime context, which can be perturbed by changing some of the $(dimension, ordinate)$ pairs defining it. Conceptually, all variables and expressions var in *all* possible dimensions. In practice, TransLucid is designed so that only a finite set of dimensions may affect any specific computation.

We introduce TransLucid with the problem of multiplying two matrices, $A_{row:m \times col:p}$ and $B_{row:p \times col:n}$, varying in dimensions col and row . The expression below defines their multiplication (*rotate* and *sum* will be defined in §2):

*UNSW Sydney, Australia; GrammaTech, USA; johnplaice@gmail.com

```

multiply.row.col  $p$   $A$   $B$ 
where
  fun multiply.dr.dc  $k$   $X$   $Y$  =  $W$ 
  where
     $\dim d \leftarrow 0$ 
     $\text{var } X' = \text{rotate.d}_c.d X$ 
     $\text{var } Y' = \text{rotate.d}_r.d Y$ 
     $\text{var } Z = X' \times Y'$ 
     $\text{var } W = \text{sum.d } k Z$ 
  end
end

```

where

- function *multiply* has *five* arguments;
- the two formal parameters d_r and d_c are *dimensional parameters*, and are introduced with a dot; these two parameters define *dimensions of variance* for X and Y ;
- the three formal parameters p , A and B are *intensional parameters*, and are introduced with a space;
- actual parameters *row*, *col*, p , A and B are passed in to *multiply*, respectively, as formal parameters d_r , d_c , k , X and Y ;
- d is an additional, temporary dimension identifier;
- X' corresponds to changing the d_r and d_c variance in X to d_r and d variance;
- Y' corresponds to changing the d_r and d_c variance in Y to d and d_c variance;
- Z is a 3-dimensional data structure corresponding to the pointwise multiplication of X' and Y' ;
- W corresponds to the collapsing through summation of the first k entries in the d direction of Z .

In the semantics for the language, in the environment providing meaning to all identifiers, each dimension identifier is mapped to a dimension, which is a natural number. When a **where** with a local dimension identifier is entered, a new dimension must be allocated. This is done by choosing the smallest dimension not currently being used.

The trickiest part of the semantics is for functions. In particular, the body of a function is evaluated with respect to a context that combines elements of the context in which the abstraction was created and the context in which the application takes place. The semantics ensures that there is never any confusion between the two.

2 Motivating examples

Like in all languages derived from ISWIM, a program in TransLucid is an expression. All expressions in TransLucid are manipulated in an arbitrary-dimensional *context*, which corresponds to an index in the Cartesian coordinate system. As an expression is evaluated, the context may be *queried*, dimension by dimension, in order to produce an answer. In so doing, other expressions may need to be evaluated in other contexts.

2.1 Constants

The simplest expression in TransLucid is the *constant*. If we consider the value 42, its value is 42, whatever the context. Below, we show the variance of 42 if we allow dimension δ_1 to vary:

	$\delta_1 \rightarrow$							
42	0	1	2	3	4	5	...	
	42	42	42	42	42	42	...	

What this table means is that in context $\{\delta_1 \mapsto 0\}$, i.e., where the ordinate of dimension δ_1 is the value 0, the value of expression ‘42’ is 42. The same holds true whatever the value of $j \in \mathbb{N}$, for all contexts $\{\delta_1 \mapsto j\}$.

The next example uses two dimensions. If, for example, the context is $\{\delta_0 \mapsto 0, \delta_1 \mapsto 0\}$, i.e., where both dimensions δ_0 and δ_1 take on the value 0, the value of expression ‘42’ is still 42. The same holds true for all contexts $\{\delta_0 \mapsto i, \delta_1 \mapsto j\}$, where $i, j \in \mathbb{N}$.

	$\delta_1 \rightarrow$							
42	0	1	2	3	4	5	...	
$\delta_0 \downarrow 0$	42	42	42	42	42	42	...	
1	42	42	42	42	42	42	...	
2	42	42	42	42	42	42	...	
\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\ddots	

The ‘variance’ of ‘42’ in further dimensions would still yield, of course, the same result.

2.2 Dimension queries

For it to be possible for the context to affect the result of the evaluation of an expression, the context must be *queried*, using the $\#$ operator. For expression ‘ $\#d_1$ ’, assuming that dimension identifier d_1 is mapped to dimension δ_1 , then the expression varies as δ_1 varies:

	$\delta_1 \rightarrow$							
$\#d_1$	0	1	2	3	4	5	...	
	0	1	2	3	4	5	...	

In, for example, context $\{\delta_1 \mapsto 4\}$, the value of expression ‘ $\#d_1$ ’ is 4. In fact, for every context $\{\delta_1 \mapsto j\}$, where $j \in \mathbb{N}$, the value of $\#d_1$ is j .

Below, we see the variance of ‘ $\#d_1$ ’ in two dimensions, where dimension identifier d_0 is mapped to dimension δ_0 . There we can see that no matter what the ordinate for dimension δ_0 , only the ordinate for dimension δ_1 in the current context is of relevance for evaluating expression ‘ $\#d_1$ ’. For all contexts $\{\delta_0 \mapsto i, \delta_1 \mapsto j\}$, where $i, j \in \mathbb{N}$, the value of $\#d_1$ is j .

	$\delta_1 \rightarrow$						
$\#d_1$	0	1	2	3	4	5	...
$\delta_0 \downarrow 0$	0	1	2	3	4	5	...
1	0	1	2	3	4	5	...
2	0	1	2	3	4	5	...
\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\ddots

2.3 Pointwise operators

As in all languages, TransLucid expressions allow the use of operators such as $+$ for addition, \times for multiplication, and so on. In TransLucid, these operators are applied *pointwise* to their arguments. In other words, in a given context κ , the expression $E_1 + E_2$ yields the sum of E_1 in κ and of E_2 in κ . Below is the variance of $\#d_0 + \#d_1$ in two dimensions:

	$\delta_1 \rightarrow$						
$\#d_0 + \#d_1$	0	1	2	3	4	5	...
$\delta_0 \downarrow 0$	0	1	2	3	4	5	...
1	1	2	3	4	5	6	...
2	2	3	4	5	6	7	...
\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\ddots

In context $\{\delta_0 \mapsto i, \delta_1 \mapsto j\}$, for all $i, j \in \mathbb{N}$, $\#d_0 + \#d_1$ has the value $i + j$.

2.4 Context changes

If the context can be queried in TransLucid, then it also needs to be changeable. This is done by placing a tuple in square brackets, using it to change the current context before continuing on with the evaluation of the expression. Below, the expression $\#d_0 + \#d_1$ is evaluated in a new context, which is created by incrementing dimension- δ_1 ’s ordinate by 1.

	$\delta_1 \rightarrow$						
$(\#d_0 + \#d_1) [d_1 \leftarrow \#d_1 + 1]$	0	1	2	3	4	5	...
$d_0 \downarrow 0$	1	2	3	4	5	6	...
1	2	3	4	5	6	7	...
2	3	4	5	6	7	8	...
\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\ddots

So, for all $i, j \in \mathbb{N}$, expression $(\#d_0 + \#d_1) [d_1 \leftarrow \#d_1 + 1]$ therefore evaluates to $i + j + 1$ in context $\{\delta_0 \mapsto i, \delta_1 \mapsto j\}$.

2.5 Factorial: version one

TransLucid, of course, needs variables, defined through equations. We introduce these with the factorial function, presented as a variable varying in dimension δ_1 , whose first few entries can be found below:

	$\delta_1 \rightarrow$								
<i>fact</i>	0	1	2	3	4	5	6	7	...
	1	1	2	6	24	120	720	5040	...

The definition in TransLucid is recursive, with a base case and an inductive case.

```

var fact = if #d1 ≡ 0 then 1
           else #d1 × (fact [d1 ← #d1 - 1])

```

2.6 Ackermann: version one

The Ackermann function is one of the first recursive functions discovered that is not primitive recursive. It grows so fast that in general it cannot be computed once its first argument is greater than 3. Here it is presented as a variable varying in dimensions δ_0 and δ_1 .

	$\delta_1 \rightarrow$							
<i>ack</i>	0	1	2	3	4	5	...	
$\delta_0 \downarrow 0$	1	2	3	4	5	6	...	
1	2	3	4	5	6	7	...	
2	3	5	7	9	11	13	...	
3	5	13	29	61	125	253	...	
4	13	65533	...					
5	65533	...						
⋮	⋱							

Here is the definition for Ackermann in TransLucid.

```

var ack =
  if #d0 ≡ 0 then #d1 + 1
  else if #d1 ≡ 0 then ack [d0 ← #d0 - 1, d1 ← 1]
  else ack [d0 ← #d0 - 1, d1 ← ack [d1 ← #d1 - 1]]

```

In the general (**else**) case, note that the nested context change is only changing the value for dimension δ_1 , since the value for dimension δ_0 need not be changed. This is similar to what happens with differential equations, in which only the dimensions of relevance are written down.

2.7 Standard functions

A function can carry both *value parameters* and *named parameters*. Below are some standard TransLucid functions.

```

fun index.d = #d + 1
fun first.d X = X [d ← 0]
fun next.d X = X [d ← #d + 1]
fun prev.d X = X [d ← #d - 1]
fun fby.d X Y = if #d < 1 then X else prev.d Y

```

Each of these declared functions has a single dimensional parameter d . Function *index* takes no intensional parameters, while *first*, *prev*, and *next* take one intensional parameter each, and *fby* takes two intensional parameters.

2.8 Factorial: version two

If we wish to write factorial as a function, we write it as taking a value parameter.

```

fun fact n = f
  where
    dim d ← n
    var f = 1 fby.d (index.d × f)
  end

```

It uses a local dimension identifier d , whose corresponding dimension δ 's ordinate is initially set to n . The stream f varies in dimension d . Note that *index.d* increments the δ -ordinate, while the second argument of *fby.d* decrements the δ -ordinate, so the two cancel each other out, yielding $\#d$.

To simplify the discussion below, for a dimension identifier d , which has been mapped to dimension δ , we will interchangeably use the terms d -ordinate and δ -ordinate.

2.9 Ackermann: version two

Ackermann takes two value parameters, and is defined using two local dimensions.

```

fun ack m n = f
  where
    dim dm ← m
    dim dn ← n
    var f = next.dn(index.dn) fby.dm
              ((next.dn f) fby.dn (f [dn ← next.dm f]))
  end

```

Note the elimination of explicit manipulation of dimensions.

2.10 Sieve of Eratosthenes

The sieve of Eratosthenes generates a stream in dimension d of the prime numbers. It is built using a local dimension d' . The first line is the naturals ≥ 2 , and subsequent lines are the previous line without the multiples of the first element of the previous line.

```

fun sieve.d = f
where
  dim d' ← 0
  var f = (#d' + 2) fby.d
          (f wvr.d' (f mod (first.d' f) ≠ 0))
end

```

The function $wvr.d$ is a *filter* in the d dimension. It returns a stream in the d dimension that retains elements of the X input when the corresponding Y element is true.

```

fun wvr.d X Y =
  if first.d Y
  then X fby.d ((next.d X) wvr.d (next.d Y))
  else (next.d X) wvr.d (next.d Y)

```

Each row corresponds to the removal of the multiples of the first element in the previous row. The sequence of primes is formed by the first column.

		$d' \rightarrow$								
f		0	1	2	3	4	5	6	7	...
$d \downarrow$	0	2	3	4	5	6	7	8	9	...
	1	3	5	7	9	11	13	15	17	...
	2	5	7	11	13	17	19	23	25	...
	3	7	11	13	17	19	23	29	31	...
	⋮	⋮	⋮	⋮	⋮	⋮	⋮	⋮	⋮	⋮

2.11 Matrix multiplication

Here is the matrix multiplication example from the introduction:

```

fun multiply.dr.dc k X Y = W
where
  dim d ← 0
  var X' = rotate.dc.d X
  var Y' = rotate.dr.d Y
  var Z = X' × Y'
  var W = sum.d k Z
end

```

Function $rotate.d_1.d_2$ changes variance in dimension d_1 to variance in dimension d_2 , while function $sum.d_x.n$ X adds up the first n elements in direction d_x of stream X .

```

fun rotate.d1.d2 X = X [d1 ← #d2]
fun sum.dx n X = Y [dx ← n]
where
  var Y = 0 fby.dx (X + Y)
end

```

2.12 Tournament computation

The factorial function was first presented in an iterative manner. It can also be computed in a *tournament computation* style, in which computation takes place in a series of stages, each stage applying the same operation upon pairs of adjacent elements.

For example, for the factorial function, if we let $n = 6$, computation proceeds as in the following table. The d_2 dimension corresponds to the successive stages. Initially, the sequence from 1 to 6 is embedded in a sea of 1s. Each element of the next stage is the result of the multiplication of two elements from the preceding stage. The result is, of course, 720.

		$d_1 \rightarrow$									
		0	1	2	3	4	5	6	7	8	...
$d_2 \downarrow$	0	1	1	2	3	4	5	6	1	1	...
	1	1	6	20	6	1	1	...			
	2	6	120	1	1	...					
	3	720	1	1	...						

To illustrate how this will take place in TransLucid, we use a general higher-order function called *tournament*:

```

fun fact n = tournament n 1 index mul
where
  fun tournament n val generate f = F
  where
    dim d1 ← 0
    dim d2 ← ilog n
    var F = (default.d1 n val generate) fby.d2 (f (lChild.d1 F) (rChild.d1 F))
  end
  fun mul X Y = X × Y
  fun lChild.d X = X [d ← #d × 2]
  fun rChild.d X = X [d ← #d × 2 + 1]
  fun default.d n val generate = if #d = 0 ∨ #d > n then val else generate.d
  end

```

We explain below:

- Functions *lChild* and *rChild* reach the appropriate element from the previous stage.
- Function *default* produces a sequence in the d direction. Elements from $\{d \mapsto 1\}$ to $\{d \mapsto n\}$ are created by the *generate* function passed in as argument. Elements $\{d \mapsto 0\}$ and $\{d \mapsto m\}$, $m > n$, are all set to the default value *val*.
- Function *tournament* uses two local dimensions, d_1 and d_2 , to set up a tournament.

2.13 Families of functions

In this section, we present two approaches to computing the power function. In the first, *power* takes two arguments, n and x , and looks up the n -th element in the sequence P .

```

power n x = P
where
  dim d ← n
  var P = 1 cby.d ((lParent.d P × lParent.d P) alt.d (x × prev.d P))
end

```

The *power* function uses the following three auxiliary functions:

```

fun lParent.d X = X [d ← #d/2]
fun cby.d X Y = if #d < 1 then X else Y
fun alt.d X Y = if #d mod 2 = 0 then X else Y

```

Another approach is to create a sequence of functions, i.e., a *family* of functions.

```

power' n = P
where
  dim d ← n
  var P' = (fn x → 1) cby.d ((fn x → lParent.d P' x × lParent.d P' x) alt.d
    (fn x → x × prev.d P' x))
end

```

Variable P' is a sequence of functions, varying in the d dimension.

2.14 Families of filters

The next example is a family of filters. The function *nnext* takes the intensional parameter n as argument.

```

nnext n = N
where
  dim dn ← n
  var N = (fn.d X → X) fby.dn (fn.d X → next.d (N.d X))
end

```

It returns the n -th element in the d -sequence

$$N = \langle id, next.d, next.d^2, \dots, next.d^n, \dots \rangle.$$

2.15 Taylor series expansion

Our last example is to compute the Taylor series expansion for a function f around point a for point x .

$$\sum_{n=0}^{\infty} \frac{f^{(n)}(a)}{n!} (x - a)^n$$

Function *taylor* takes as input a stream *derivs* of derivatives of f at point a in direction d .

```

fun taylor.d a x derivs = F
where
  var F = sum.d (index.d) G
  var G = derivs / (fact (#d)) × (pow (#d) (x - a))
end

```

The Taylor series expansion for the sine function around integral multiples of 2π yields:

```

taylor.d a x sinderivs
where
  var sinderivs =
    0 fby.d (1 fby.d (0 fby.d (-1 fby.d sinderivs)))
end

```

2.16 On intensionality

The fact that a variable, conceptually varies in all dimensions forces the programmer to think in a *intensional* manner, as it is in general not possible, nor desirable, to enumerate all of these dimensions. For example, the phrase “*Five degrees less than yesterday’s temperature*” could be written as the expression

$$(t - 5) [date \leftarrow \#date - 1]$$

where t is the temperature and dimension *date* keeps track of the current date. Note that this expression does not make explicit where t is to be evaluated, which might vary according to other dimensions, such as *latitude*, *longitude*, *altitude*, *timeOfDay*, and so forth. This approach is consistent with the use of dimensions in differential dimensions, where only relevant dimensions are written down.

3 Semantics

3.1 Notation for function manipulation

Definition 1. Let S be a set and $\perp \notin S$. We define $S_\perp = S \cup \{\perp\}$. The trivial order \sqsubseteq over S_\perp is given by

$$s \sqsubseteq s' \quad \text{iff} \quad s = \perp \vee s = s'.$$

Definition 2. Let A and B be two sets. A total function f from A to B is written $f : A \rightarrow B$. If \sqsubseteq is defined over B , then the order \sqsubseteq over $A \rightarrow B$ is given by

$$f \sqsubseteq f' \quad \text{iff} \quad \forall a \in A, f(a) \sqsubseteq f'(a).$$

Definition 3. Let A and B be two sets. A partial function f from A to B is written $f : A \mapsto B$.

When f is a function with finite domain $\{v_i\}_{i \in m}$, we will write it as its graph

$$\{v_i \mapsto f(v_i)\}_{i \in m}.$$

When the graph is empty, we will write it as \emptyset .

Definition 4. Let $f : A \rightarrow B$ and $g : A \mapsto B$. The perturbation of f by g , written $f \dagger g$, is defined by:

$$(f \dagger g)(v) = \begin{cases} g(v), & v \in \text{dom } g \\ f(v), & \text{otherwise.} \end{cases}$$

Definition 5. Let $f : A \rightarrow B$, and let $S \subseteq A$. The domain restriction of f to S , written $f \upharpoonright S$, is defined by

$$(f \upharpoonright S)(v) = \begin{cases} f(v), & v \in S. \end{cases}$$

3.2 Syntax

Definition 6. The identifiers ($s \in S$) that can appear in expressions are:

- constant symbols ($c \in C$),
- dimension identifiers ($d \in D$),
- variable (and function) identifiers ($x \in X$),

where C , D and X are mutually disjoint and $S = C \cup D \cup X$.

Definition 7. The abstract syntax for an expression $E \in \mathbf{Expr}$ is as follows:

$E ::= x$		variable ($x \in X$)
$c(E, \dots)$		constant ($c \in C$)
$\#d$		context lookup ($d \in D$)
$E [d \leftarrow E, \dots]$		context perturbation
$E.(d, \dots)(E, \dots)$		function application
$\mathbf{fn}.(d, \dots)(x, \dots) \rightarrow E$		function abstraction
$E \mathbf{where}$		local declarations
$\mathbf{dim} \ d \leftarrow E, \dots$		dimensions
$\mathbf{var} \ x = E, \dots$		variables
$\mathbf{fun} \ x.(d, \dots)(x, \dots) = E, \dots$		functions
end		

3.3 Semantics

Definition 8. Let $V \supseteq \{\mathit{true}, \mathit{false}\}$ be a set of atomic values, where $\perp \notin V$. The domains defined over V are:

$$\mathbf{V} = V_{\perp} \tag{1}$$

$$\mathbf{A} = \bigcup_{m \in \omega} (\mathbf{V}^m \rightarrow \mathbf{V}) \tag{2}$$

$$\mathbf{K} = \mathbb{N} \rightarrow \mathbf{V} \tag{3}$$

$$\mathbf{H} = \mathbf{K} \rightarrow (\mathbf{V} \cup \mathbf{F}) \tag{4}$$

$$\mathbf{F} = \bigcup_{m, n \in \omega} \mathbb{N}^m \times \mathbf{H}^n \rightarrow \mathbf{H} \tag{5}$$

$$\mathbf{Env} = (C \rightarrow \mathbf{A}) \cup (D \rightarrow \mathbb{N}_{\perp}) \cup (X \rightarrow \mathbf{H}) \tag{6}$$

- \mathbf{V} ($\ni v$) is the set of basic values.
- \mathbf{A} ($\ni op$) is the set of atomic operators of arity $m \in \omega$. This set will typically include the standard arithmetic and comparison operators, as well as the `if-then-else` operator.
- \mathbf{K} ($\ni \kappa$) is the set of contexts. Elements of the domain of a context are called dimensions. Elements of the range of a context are called ordinates. Thus a context consists of a set of (dimension, ordinate) pairs.
- \mathbf{H} ($\ni \eta$) is the set of intensions, mapping contexts to values.
- \mathbf{F} ($\ni f$) is the set of indexed functions of dimension arity m and intension arity n .
- $\mathbf{V} \cup \mathbf{F}$ is the set of first-class values, which are manipulated by expressions.
- \mathbf{Env} is the set of environments.
- $\perp_{\mathbf{H}} = \lambda \kappa. \perp$ is the completely undefined intension.
- The order \sqsubseteq over these domains is given using Definitions 1–2.

Definition 9. For $\kappa \in \mathbf{K}$, $\text{dims}(\kappa)$ is the set of active dimensions of κ , i.e.,

$$\text{dims}(\kappa) = \{\delta \in \mathbb{N} \mid \kappa(\delta) \neq \perp\}.$$

The set \mathbf{K}^f consists of all those $\kappa \in \mathbf{K}$ that are finite-dimensional, i.e.,

$$\text{card}(\text{dims}(\kappa)) < \infty.$$

For $\kappa \in \mathbf{K}^f$, $\text{rk}(\kappa)$ is the least dimension greater than all of the active dimensions of κ , i.e.,

$$\forall \delta \in \text{dims}(\kappa), \delta < \text{rk}(\kappa).$$

Definition 10. The set \mathbf{H}^m consists of all those $\eta \in \mathbf{H}$ that are monotonic, i.e.,

$$\forall \kappa, \kappa', \kappa \sqsubseteq \kappa' \implies \eta(\kappa) \sqsubseteq \eta(\kappa').$$

For $\eta \in \mathbf{H}^m$, $\text{minctxts}(\eta)$ is the set of minimal contexts for η , i.e.,

$$\text{minctxts}(\eta) = \{\kappa \in \mathbf{K} \mid \eta(\kappa) \neq \perp \wedge \forall \kappa' \sqsubset \kappa, \eta(\kappa') = \perp\}.$$

Definition 11. For $\eta \in \mathbf{H}^m$, $\text{dims}(\eta)$ is the set of active dimensions of η , i.e.,

$$\text{dims}(\eta) = \bigcup \{\text{dims}(\kappa) \mid \kappa \in \text{minctxts}(\eta)\}.$$

The set \mathbf{H}^f consists of all those $\eta \in \mathbf{H}^m$ that are finite-dimensional, i.e.,

$$\text{card}(\text{dims}(\eta)) < \infty.$$

For $\eta \in \mathbf{H}^f$, $\text{rk}(\eta)$ is the least dimension greater than all of the active dimensions of η , i.e.,

$$\forall \delta \in \text{dims}(\eta), \delta < \text{rk}(\eta).$$

Definition 12. For $\zeta \in \mathbf{Env}$, $\text{dims}(\zeta)$ is the set of active dimensions of ζ , i.e.,

$$\text{dims}(\zeta) = \{\zeta(d) \mid d \in D \wedge \zeta(d) \neq \perp\} \cup \bigcup_{\eta \in \{\zeta(x) \mid x \in X\}} \text{dims}(\eta).$$

The set \mathbf{Env}^f consists of all those $\eta \in \mathbf{Env}$ that are finite-dimensional, i.e.,

$$\text{card}(\text{dims}(\zeta)) < \infty.$$

For $\zeta \in \mathbf{Env}^f$, $\text{rk}(\zeta)$ is the least dimension greater than all of the active dimensions of ζ , i.e.,

$$\forall \delta \in \text{dims}(\zeta), \delta < \text{rk}(\zeta).$$

Definition 13. Let $E \in \mathbf{Expr}$, $\zeta \in \mathbf{Env}^f$, and $\kappa \in \mathbf{K}^f$. Then the semantics for E with respect to ζ and κ is given by

$$\llbracket E \rrbracket \zeta \kappa,$$

where the rules for

$$\llbracket \cdot \rrbracket : \mathbf{Expr} \rightarrow \mathbf{Env}^f \rightarrow \mathbf{K}^f \rightarrow \mathbf{V} \quad (7)$$

are given in Equations 8–14.

$$\llbracket x \rrbracket \zeta \kappa = \zeta(x)(\kappa) \quad (8)$$

$$\llbracket c(E_i)_{i \in m} \rrbracket \zeta \kappa = \zeta(c)(\llbracket E_i \rrbracket \zeta \kappa)_{i \in m} \quad (9)$$

$$\llbracket \#d \rrbracket \zeta \kappa = \kappa(\zeta(d)) \quad (10)$$

$$\llbracket E @ [d_i \leftarrow E_i]_{i \in m} \rrbracket \zeta \kappa = \llbracket E \rrbracket \zeta(\kappa \dagger \{\zeta(d_i) \mapsto \llbracket E_i \rrbracket \zeta \kappa\}_{i \in m}) \quad (11)$$

$$\left[\begin{array}{l} E \text{ where} \\ \dim_{i \in p} \quad d_i \leftarrow E_i \\ \text{var}_{j \in p+1..q} \quad x_j = E_j \\ \text{fun}_{k \in q+1..r} \quad x_k.(d_{i,k})_{i \in m_k}(x_{j,k})_{j \in n_k} = E_k \\ \text{end} \end{array} \right] \zeta \kappa \quad (12)$$

$$= \text{let } \nu = \max(\text{rk}(\zeta), \text{rk}(\kappa))$$

$$\delta_i = \nu + i, \quad i \in p$$

$$\kappa' = \kappa \dagger \{\delta_i \mapsto \llbracket E_i \rrbracket \zeta \kappa\}_{i \in p}$$

$$\zeta_0 = \zeta \dagger \{d_i \mapsto \delta_i\}_{i \in p}$$

$$\dagger \{x_j \mapsto \perp_{\mathbf{H}}\}_{j \in p+1..q}$$

$$\dagger \{x_k \mapsto \perp_{\mathbb{N}^{m_k} \times \mathbf{H}^{n_k} \rightarrow \mathbf{H}}\}_{k \in q+1..r}$$

$$\zeta_{\alpha+1} = \zeta \dagger \{d_i \mapsto \delta_i\}_{i \in p}$$

$$\dagger \{x_j \mapsto \llbracket E_j \rrbracket \zeta_\alpha\}_{j \in p+1..q}$$

$$\dagger \{x_k \mapsto \llbracket \text{fn}.(d_{i,k})_{i \in m_k}(x_{j,k})_{j \in n_k} \rightarrow E_k \rrbracket \zeta_\alpha\}_{k \in q+1..r}$$

$$\zeta' = \lim_{\alpha \rightarrow \infty} \zeta_\alpha$$

$$\text{in } \llbracket E \rrbracket \zeta' \kappa'$$

$$\llbracket E.(d'_i)_{i \in m}(E'_j)_{j \in n} \rrbracket \zeta \kappa = (\llbracket E \rrbracket \zeta \kappa) (\zeta(d'_i))_{i \in m} (\llbracket E'_j \rrbracket \zeta)_{j \in n} (\kappa) \quad (13)$$

$$\llbracket \mathbf{fn}.(d_i)_{i \in m}(x_j)_{j \in n} \rrbracket \zeta \kappa = \lambda(\delta_i)_{i \in m} . (\eta_j)_{j \in n} . \kappa' \quad (14)$$

$$\text{let } \nu = \max(\text{rk}(\zeta), \text{rk}(\kappa), \text{rk}(\kappa'), \delta_i, \text{rk}(\eta_j))_{i \in m, j \in n}$$

$$\delta'_i = \nu + i, \quad i \in m$$

$$\kappa'' = \kappa \dagger \{ \delta'_i \mapsto \kappa'(\delta_i) \}_{i \in m}$$

$$\zeta' = \zeta \dagger \{ d_i \mapsto \delta'_i \}_{i \in m}$$

$$\dagger \{ x_j \mapsto \lambda \kappa_j . \eta_j (\kappa' \dagger \{ \delta_i \mapsto \kappa_j(\delta'_i) \}_{i \in m}) \}_{j \in n}$$

$$\text{in } \llbracket E \rrbracket \zeta' \kappa''$$

We explain all of the different cases for Equations 8–14 below.

- (8) *Variable*: identifier x is looked up in environment ζ to produce an intension η , which is applied to the context κ to produce a value v .
- (9) *Constant*: each expression E_i is evaluated with respect to environment ζ and context κ to produce an atomic value v_i . Constant symbol c is looked up in environment ζ to produce an atomic function op , which is then applied to the v_i to produce an atomic value v . Should $m = 0$, the value is op itself.
- (10) *Context lookup*: dimension identifier d is mapped to dimension $\delta = \zeta(d)$. The ordinate for δ is $\kappa(\delta)$.
- (11) *Context perturbation*: each expression E_i is evaluated with respect to environment ζ and context κ to produce atomic value v_i . Context κ is then perturbed by changing, for each i , the ordinate for dimension $\delta_i = \zeta(d_i)$ to v_i . Expression E is then evaluated in this new context.
- (12) *Local declarations*: we give the semantics for an expression of the form:

E where

$$\mathbf{dim}_{i \in p} \quad d_i \leftarrow E_i$$

$$\mathbf{var}_{j \in p+1..q} \quad x_j = E_j$$

$$\mathbf{fun}_{k \in q+1..r} \quad x_k . (d_{i,k})_{i \in m_k} (x_{j,k})_{j \in n_k} = E_k$$

end

Given an environment ζ and a context κ , expression E must be evaluated with respect to a new environment ζ' and a new context κ' .

For each of the new dimension identifiers d_i , a new dimension δ_i needs to be allocated. The first freely available dimension ν is the maximum of the ranks of ζ and κ . For

dimension identifier d_i , the new dimension is $\delta_i = \nu + i$. The new context κ' perturbs κ by mapping δ_i to $\llbracket E_i \rrbracket \zeta \kappa$.

A sequence of environments ζ_α , $\alpha \in \mathbb{N}$, is initialized by creating environment

$$\begin{aligned} \zeta_0 &= \zeta \dagger \{d_i \mapsto \delta_i\}_{i \in p} \\ &\quad \dagger \{x_j \mapsto \perp_{\mathbf{H}}\}_{j \in p+1..q} \\ &\quad \dagger \{x_k \mapsto \perp_{\mathbb{N}^{m_k} \times \mathbf{H}^{n_k} \rightarrow \mathbf{H}}\}_{k \in q+1..r}. \end{aligned}$$

Environment $\zeta_{\alpha+1}$ is created from the meanings of the variables x_j in environment ζ_α :

$$\begin{aligned} \zeta_{\alpha+1} &= \zeta \dagger \{d_i \mapsto \delta_i\}_{i \in p} \\ &\quad \dagger \{x_j \mapsto \llbracket E_j \rrbracket \zeta_\alpha\}_{j \in p+1..q} \\ &\quad \dagger \{x_k \mapsto \llbracket \mathbf{fn}.(d_{i,k})_{i \in m_k} (x_{j,k})_{j \in n_k} \rightarrow E_k \rrbracket \zeta_\alpha\}_{k \in q+1..r}. \end{aligned}$$

Environment ζ' is the limit of the sequence of ζ_α :

$$\zeta' = \lim_{\alpha \rightarrow \infty} \zeta_\alpha.$$

Finally, expression E is evaluated with respect to environment ζ' and context κ' to produce a value v .

- (13) *Function application*: a function f is produced by evaluating expression E with respect to ζ and κ . For each i , environment ζ is applied to actual dimensional parameter d_i to produce dimension δ_i . For each j , the meaning of actual expression parameter E_j is applied to ζ to produce intension η_j . Then, function f is applied to the dimensional actual parameters δ_i , the intensional actual parameters η_j , and the abstraction context κ .
- (14) *Function abstraction*: the evaluation of an abstraction expression takes place in an environment ζ and an *abstraction context* κ . The result is an abstraction that takes as input a set of dimensional actual parameters δ_i , a set of intensional actual parameters η_j , and an *application context* κ' . The technical difficulty to be resolved is the interaction between κ and κ' in the evaluation of the function body E .

For each formal dimensional parameter d_i , a new dimension δ'_i needs to be allocated. The first freely available dimension ν is the maximum of the ranks of ζ , κ and κ' , the ranks of each of the η_j , and the δ_i values themselves. For each i , the new dimension is $\delta'_i = \nu + i$.

The context κ'' of evaluation of the body E is the abstraction context κ , perturbed by mapping, for each i , the new dimension δ'_i to the ordinate of the actual dimensional parameter δ_i in application context κ' :

$$\kappa'' = \kappa \dagger \{\delta'_i \mapsto \kappa'(\delta_i)\}_{i \in m}.$$

As E is being evaluated, κ'' will possibly be modified, and it is important that when a variable x is encountered, that it be evaluated to the appropriate context. To make this possible, the environment ζ' for evaluating E is given below:

$$\begin{aligned}\zeta' &= \zeta \dagger \{d_i \mapsto \delta'_i\}_{i \in m} \\ &\quad \dagger \{x_j \mapsto \lambda \kappa_j. \eta_j(\kappa' \dagger \{\delta_i \mapsto \kappa_j(\delta'_i)\}_{i \in m})\}_{j \in n}.\end{aligned}$$

Each formal dimensional parameter d_i is mapped to the new dimension δ'_i . Each formal intensional parameter x_j is mapped to an intension, which works as follows: should x_j need to be evaluated, it will be evaluated with respect to the *application context* κ , taking into account changes that have been made to the ordinates of the new dimensions during the evaluation up to that point. If κ_j is the context encountered when x_j is to be evaluated, the value needed is precisely

$$\eta_j(\kappa' \dagger \{\delta_i \mapsto \kappa_j(\delta'_i)\}_{i \in m})_{j \in n}.$$

In all other cases, a variable x needs to be evaluated with respect to the *abstraction context*, taking into account changes that have been made to the ordinates of that context during the evaluation up to that point. The use of κ'' also works here, since the new dimensions are all greater than the dimensions needed to evaluate x .

If we let $m = 0$, then Equations 13–14 become:

$$\llbracket E(E'_j)_{j \in n} \rrbracket \zeta \kappa = (\llbracket E \rrbracket \zeta \kappa) (\llbracket E'_j \rrbracket \zeta)_{j \in n}(\kappa) \quad (15)$$

$$\llbracket \mathbf{fn}.(x_j)_{j \in n} \rrbracket \zeta \kappa = \lambda(\eta_j)_{j \in n}.\kappa'. \quad (16)$$

$$\text{let } \zeta' = \zeta \dagger \{x_j \mapsto \lambda \kappa_j. \eta_j \kappa'\}_{j \in n}$$

$$\text{in } \llbracket E \rrbracket \zeta' \kappa$$

Note that in Equation 16, intension η_j is always applied to κ' , and since $\eta_j(\kappa)$ is constant, it can be computed before entering the function. Equations 15–16 can thus be rewritten as:

$$\llbracket E(E'_j)_{j \in n} \rrbracket \zeta \kappa = (\llbracket E \rrbracket \zeta \kappa) (\llbracket E'_j \rrbracket \zeta \kappa)_{j \in n} \quad (17)$$

$$\llbracket \mathbf{fn}.(x_j)_{j \in n} \rrbracket \zeta \kappa = \lambda(v_j)_{j \in n}. \quad (18)$$

$$\text{let } \zeta' = \zeta \dagger \{x_j \mapsto v_j\}_{j \in n}$$

$$\text{in } \llbracket E \rrbracket \zeta' \kappa$$

So in the special case where there are no dimensional parameters, a TransLucid function is simply a normal function in a standard functional language.

4 Future work

In lieu of a conclusion, we focus on some possible future work.

The TransLucid language presented in this paper is a functional language manipulating intensions, which map contexts to values. Since this is a functional language, these values can include functions.

However, most modern functional languages allow the use of partially applied functions, using currfication. For the moment, TransLucid does not support currfication, for the following reason. Consider a function of the form

$$\text{fun } f.d \ X = E$$

and the possibility of computing $g = f.d_1$ and then later applying g to an argument E_2 . It could be possible that the way the calls have been set up, that dimension identifier d_1 would not be available in the environment at the application point $g \ E_2$. To do this would require some kind of static semantics, i.e., a type checker. We leave this as future work.

The denotational semantics of TransLucid leads naturally to an implementation, although not necessarily to an efficient one. For recursively defined structures, it is common to use memoization, i.e., caching, to ensure that previously computed values do not need to be recomputed.

Memoization for variables, indexed by the current context, is a promising possibility. The difficulty, like for the semantics, lies in the implementation of functions with memoization. When the body of a function would be evaluated, and an intensional parameter encountered, it is important to keep track of the different possible invocations of the function. This can be done by hashing the expressions that are passed in as arguments. Once again, we leave this as future work.